

Koden beskrevet her kan bruges til at skrive et tal i et 7-segment.

Det sker i et funktionskald, hvortil der medbringes det tal, der skal skrives.

De 7 enkelte segmenter, der udgør et tal i et 7-segment, er beskrevet i en tabel. I et ARRAY, med 10 elementer, én byte for hver tal-mønster.

I et Array har det første element nummer "0"

Et "1" angiver, at segmentet skal tændes.

Ligeledes er de pin-numre, der er tilsluttet de enkelte segmenter i 7-segmentet angivet i et ARRAY. Det betyder, at de ikke behøves at komme i en bestemt rækkefølge.

Det er her lavet sådan, at loop øger et tal fra 0 og op til 9 hvert sekund.

Loop kalder så sub-funktionen, og medsender det tal, der skal vises.

Funktionen "skriv7seg."

Når funktionen kaldes, medfølger det tal, der skal skrives, som en parameter.

For nemheds skyld, hentes først 7-seg-mønsteret ind i variabelen x, for det tal, der skal vises.

```
x = tabel[x];
```

Hvis det medbragte tal fx er 8, hentes mønsteret fra plads 8 i mønstertabellen ind i variabelen "x".

Hver bit i variabelen x beskriver nu med et «1» hvilke segmenter i 7-segmentet, der skal lyse.

Alle 7 bit skal sendes ud til de respektive pins. Det sker i en for-løkke, der "kører" 7 gange, 1 for hver segment.

En digitalWrite instruktion sender nu de respektive bits ud til hver sin pin. I første omgang er "i" = 0. Dvs. at der skrives til den pin, der er beskrevet på plads 0 i ledPin-Arrayet.

```
digitalWrite((ledPin[i]), bitRead(x, i));
```

Værdien, der udskrives ved at læse værdien af bit 0 i variabelen "x".

Hvis "i" er nul, vil ovenstående kodelinje være det samme som

```
digitalWrite(3,1);
```

Man behøver dog ikke omvejen omkring at indlæse 7-segment-mønsteret i x.

Følgende kode kan også bruges !

```
digitalWrite(ledPin[i], bitRead(tabel[x], i));
```

Men så skal kodelinjen `x = tabel[x];` fjernes.

```
//*****'
```

```
/*
   Kode til styring af 1 stk 7-segment
   Lagt ud i et funktionskald.

   April 22/4 - 2021
   Redigeret 2/06-2025

   / Valle

*/

uint8_t tabel[10] = {
    /* Mønstertabel, hvilke segmenter skal tændes
    for at vise hvilket tal ?

        a
    f  b
        g
    e  c
        d

    *gfedcba, skal være i denne rækkefølge !! */

    B00111111, // 0
    B00000110, // 1
    B01011011, // 2
    B01001111, // 3
    B01100110, // 4
    B01101101, // 5
    B01111101, // 6
    B00000111, // 7
    B01111111, // 8
    B01101111 // 9
};

// def af Pin-Array. Hvilke segmenter
// styres af hvilke pins ??
// pin til a, b, c, d, e, f, g
uint8_t ledPin[] = { 3, 2, 4, 9, 10, 12, 11 };
```

```
// def af tæller
uint8_t taeller = 0;
uint8_t x;

//*****

void setup() { // def som output

    for (uint8_t x = 0; x < 7; x++) { // Det giver 7
        pinMode(ledPin[x], OUTPUT);
    }
} // EndSetup

// *****

void loop() {
    skriv7seg(taeller);
    delay(1000);
    taeller++;
    if (taeller > 9) taeller = 0;
} // EndLoop

// *****

void skriv7seg(uint8_t x) {
    x = tabel[x]; // hent tallets mønster
    for (uint8_t i = 0; i < 7; i++) {
        // skriv til pin, def i array,
        // bit nummer "i" i variablen x
        digitalWrite((ledPin[i]), bitRead(x, i));
    } // endFor
} // endskriv7seg

//*****
```